

# RPG /Free Format Enhancements since 7.1

Matt Spies  
@matt\_spies

# Why go toward /Free format RPG?

- Self documenting code
- Easier to code and to maintain
- More horizontal space to code

# How we got here?

- V5R1 - /Free format C Specifications
- V5R3 - /Free format SQL
- 7.1 - /Free format H, F, D, P Specifications
- Requirements
  - 5770WDS-SI51094 for RPG compiler (approx. 10MB)
  - Superseded by 5770WDS-SI52307
  - DB2 Group PTF SF99701 Level 26 for SQL precompiler support

# Ground Rules-Enhanced RPG Compiler

- /Free and /End-Free are no longer needed.
- I & O Specs and RPG Cycle remain Fixed Format.
- F & D Specs can be mixed together.
- Free Format and Fixed Format can be intermixed... within the standard coding guidelines, of course!
- Free Format has sensible defaults.
- Op Codes are not case specific.
- Free Format code must be between columns 8-80.
- Semicolon is necessary at the end of every free format statement ;

# Source Entry Utility (SEU) EDITORS

Rational Developer for i (RDi) 9.0.1

```
Columns . . . : 6 76      Edit      MSSLIB/QRPGLSRC
SEU=>          ITEMBL102F
FMT ** ... 1 ...+... 2 ...+... 3 ...+... 4 ...+... 5 ...+... 6 ...+... 7 ...+
***** Beginning of data *****
0001.00  ctl-opt DftName( ITEMBL102F )
0002.00      Option( *NoDebugIO: *SrcStmt )
0003.00      DftActGrp( *No )
0004.00      ActGrp( 'QILE' );
0005.00
0006.00  // *****
0007.00  // * Files
0008.00  // *****
0009.00
0010.00  dcl-f S1Qnty Disk
0011.00      Usage(*input);
0012.00
0013.00  dcl-f ItemBl Disk
0014.00      Usage(*input) Keyed
0015.00      Prefix(BL_);
0016.00

F3=Exit  F4=Prompt  F5=Refresh  F9=Retrieve  F10=Cursor  F11=Toggle
F16=Repeat find  F17=Repeat change  F24=More keys
(C) COPYRIGHT IBM CORP. 1981, 2007.
M@  B 02/009
```

```
ITEMBL102F.RPGLE
Line 1      Column 1      Replace
..... Free-Form+++++++
000100      ctl-opt DftName( ITEMBL102F )
000200          Option( *NoDebugIO: *SrcStmt )
000300          DftActGrp( *No )
000400          ActGrp( 'QILE' );
000500
000600      // *****
000700      // * Files
000800      // *****
000900
001000      dcl-f S1Qnty Disk
001100          Usage(*input);
001200
001300      dcl-f ItemBl Disk
001400          Usage(*input) Keyed
001500          Prefix(BL_);
001600
001700      dcl-f TMP113 Disk
001800          Usage(*output);
001900
002000      // *****
002100      // * Prototypes
002200      // *****
002300
002400      dcl-pr Itembl102F Zoned(7:0) end-pr;
002500
002600      dcl-pi *n;
002700          dcl-parm NxtCycleCYMD Zoned(7:0);
002800      end-pi;
002900
003000      // *****
003100      // * Variables
003200      // *****
```

# H Spec

## Fixed Format

```
H NOMAIN  
H OPTION( *NODEBUGIO: *SRCSTMT)  
H DATFMT( *MDY )  
H DFTACTGRP( *NO )  
H BNDDIR( 'SERVICELIB/SERVICEDIR')
```

## /Free Format

```
Ctl-Opt NOMAIN  
    OPTION( *NODEBUGIO: *SRCSTMT)  
    DATFMT( *MDY )  
    BNDDIR( 'SERVICELIB/SERVICEDIR' ) ;  
  
// DFTACTGRP(*NO) not needed if  
// ACTGRP, BNDDIR, or STGMDDL  
// keywords exist.
```

# F Spec

## Fixed Format

```
F InpFile      IF      E          DISK
F UpdFile      UF      E          K    DISK
F OutFile      O      A E          DISK
F UpdOutFileUF A      E          K    DISK
F PrintFile    O          F    132    PRINTER Of1Ind(*INOF)
F DspFile      CF      E          WORKSTN
* U is used for UPDATE & DELETE
```

# F Spec

## /Free Format

```
Dcl-F InpFile; // Usage(*Input) implied.  
           // Disk or Disk(*Ext) implied.  
Dcl-F UpdFile Usage(*Update) Keyed;  
Dcl-F OutFile Usage(*Output);  
Dcl-F UpdOutFile Usage(*Output: *Update) Keyed;  
Dcl-F PrintFile Printer(132) Of1Ind(*INOF);  
Dcl-F DspFile Workstn;  
// Usage(*Delete) is necessary for delete  
// Usage(*Update) implies *Input and *Update
```



# D Spec - Variables

## Fixed Format

D Header	C			'FASUG'
D Title	S	60A		
D ISODate	S		D	DatFmt(*ISO)
D Index	S	3I 0		Inz(1)
D ShipQuantity	S	7S 0		
D SalesAmount	S	10P 2		
D FoundItem	S	N		Inz(*Off)
D Counter				Like(Index)

# D Spec - Variables

## /Free Format

```
Dcl-C Header `FASUG` ; // Const(`FASUG`) optional
Dcl-S Title Char(60);
Dcl-S ISODate Date(*ISO); // No DatFmt Keyword
Dcl-S Index Int(3) Inz(1); // 3, 5, 10, 20 digits
Dcl-S ShipQuantity Zoned(7); // or Zoned(7:0)
Dcl-S SalesAmount Packed(10:2);
Dcl-S FoundItem Ind Inz(*Off); // No more 15 character
Dcl-S Counter Like(Index); // limit on field names!
```

# D Spec - Data Structures

## Fixed Format

D	InputRecDS	DS			LikeRec(RcdFmt)
D	OutputRecDS	DS			LikeDS(InputRecDS)
D	CurTimeDS	DS			
D	CurTime		6S	0	
D	CurHour		2S	0	Overlay(CurTime:1)
D	CurMin		2S	0	Overlay(CurTime:3)
D	CurSec		2S	0	Overlay(CurTime:5)
D	InfDS	DS			
D	Num_Rclds		156	159I	0
D	OneLine	DS		80	

# D Spec - Data Structures

## /Free Format

```
Dcl-DS InputRecDS LikeRec(RcdFmt);           // No End-DS required
Dcl-DS OutputRecDS LikeDS(InputRecDS);       // No End-DS required

Dcl-DS CurTimeDS;           // Dcl-DS *n ;   if there is no DS name
    CurTime Zoned(6);
    CurHour Zoned(2) Pos(1); //... Pos(1) or Overlay(CurTime:1).
    CurMin  Zoned(2) Pos(3); // Dcl-SubF is optional, except when subfield = RPG opcode
    CurSec  Zoned(2) Pos(5);
End-DS; // or End-DS CurTimeDS;
// Overlay(arrayname:startposition) needed for array data structures
Dcl-DS InfDS;
    Num_Rcds Int(10) Pos(156);
End-DS;
Dcl-DS OneLine Len(80) End-DS;
```

# D Spec - Prototypes & Procedure Interfaces

## Fixed Format

D	PgmName	PR		
D			2S	0
D			9A	
D	PgmName	PI		
D	Company		2S	0
D	OrderNumber		9A	

# D Spec - Prototypes & Procedure Interfaces

## **/Free Format**

```
Dcl-Pr PgmName;
```

```
    Company Zoned(2);
```

```
    OrderNumber Char(9);
```

```
End-Pr;
```

```
Dcl-PI PgmName;
```

```
    Company Zoned(2);
```

```
    OrderNumber Char(9);
```

```
End-PI;
```

```
// Dcl-Parm is optional, except when  
// parameter = RPG opcode (select, read)
```

# P Spec

Fixed Format

/End-Free

**P** NEXTLEVEL    B

Export

/Free

// code some stuff

/End-Free

**P** NEXTLEVEL    E

/Free

/Free Format

**Dcl-Proc** NEXTLEVEL Export;

// code some stuff

**End-Proc**;

RPG

Spec

Free Format OpCode

H

Ctl-Opt

F

Dcl-F

D

Dcl-C Dcl-S Dcl-DS End-DS Dcl-Subf

Dcl-Pr End-Pr Dcl-PI End-PI Dcl-Parm

P

Dcl-Proc End-Proc



# /End-Free

- IBM Developer Works // Excellent resource
  - <https://www.ibm.com/developerworks/ibmi/library/i-ibmi-rpg-support/>
- ILE RPG Reference
  - <http://pic.dhe.ibm.com/infocenter/iserics/v7r1m0/topic/books/sc092508a.pdf>
- “The Past, Present and Future of RPG” Tim Rowe
  - <http://iprodeveloper.com/rpg-programming/past-present-and-future-rpg>
- “Four Reasons RPG Geezers Should Care About The New Free Format RPG” Paris
  - <http://www.itjungle.com/fhg/fhg021214-story01.html>

T H A N K  
Y O U